TIETOKONESIMULAATIOT SYKSY 2007
IDL-harjoitus I (Heikki Salo 18.10.07)
ratkaisut

1) π'n likiarvon määrääminen rahan heitolla

2) Haluttua jakaumaa noudattavan satunnaismuuttujan luonti

3) Määrätyn integraalin laskeminen MC-menetelmällä

1 π'n likiarvon määrääminen rahan heitolla

Luennoissa on esimerkki $\pi : n$ likiarvon määräämisestä käyttäen Buffon'in neulanheitto menetelmää. Toista sama käyttäen rahan heittoa ruudukkoon!

Oletetaan suorakulmainen ruudukko, jossa ruutujen väli on $2L$. Heitetään rahaa, jonka halkaisija on 2R. Millä todennäköisyydellä raha osuu ruudukon leikkauspisteiden päälle? Yksinkertaisuuden vuoksi oletetaan $2R < 2L$.

Laske analyyttinen lauseke leikkaustodennäköisyydelle (sisältää π'n), sekä heittokokeeseen perustuva arvio, joka lähestyy oikeaa arvoa heittojen lukumäärän N kasvaessa. Merkitään N_{hit} heittokokeessa ruudukon leikkauspisteiden päälle osuvien heittojen lukumäärää.

Laske myös analyyttinen arvio π'lle saadun likiarvon PII keskivirheelle.

Tarkista virhearvio vertaamalla sitä esim $k = 30$ heittokokeen sarjassa (kussakin N heittoa) saatujen PII:n arvojen keskivirheeseen.



RATKAISUT:

pii_kolikko.pro

```
;--------------------------------------------------------------
;pii_kolikko.pro
;esimerkki pii:n arvioimiseksi kolikkoa heittamalla
;-------------------------------------------------------------
;Heitetaan kolikkoa, jonka halkaisija on 2R,
;ruudukolle jonka ruutujen koko on 2L  (oletetaan 2R < 2L)
;Merkitaan kolikon keskipisteen paikkaa x,y
;(mitattuna mielivaltaisesta ruudukon leikkauspisteesta)
;Milla todennakoisyydella kolikko peittaa yhden ruudukon
;leikkauskohdista = hit?
;-------------------------------------------------------------
;symmetria -> voidaan tarkastella aluetta
;             0<x<L, 0<y<L
;mikali       x^2+y^2 < R^2 -> peittaa ruudukon leikkauskohdan

;arvio todennakoisyydelle =  N_hit/N
;jossa N on kaikkien heittojen lukumaara ja N_hit 'osumien maara'

;teoreettisesti tiedetaan etta odotusarvo=
;kolikon ala/ruudun ala  = pi/4 (R/L)^2

; --> arvio pii= 4*(L/R)^2 N_hit/N
;-------------------------------------------------------------

   N=1000             ;heittojen lukumaara
   x=randomu(seed,N)
   y=randomu(seed,N)
   RL=1.0                       ;kolikon sade R/L
   R2=RL^2

   hit=where((x^2+y^2) le R2, N_hit)

;HUOM: oltava tarkkana IDL tyyppimaarittelyjen kasittelyn suhteen:
;Mika olisi tulos jos kirjoittaisit 4. (=float) sijasta 4 (=integer)?
;arvio pi:n arvolle = pii

   pii=4.*N_hit/N*1./R2

;teoreettinen virhearvio
;osumistodennakoisyys p= pi/4 (R/L)^2 kussakin heitossa
;Riippumattomat heitot, binomitodennakoisyys -->
;sigma_p_teo=sqrt(p(1-p)/N)

   p= !pi/4 * (RL)^2
   sigma_p_teo=sqrt(p*(1-p)/N)

;arvio pii= 4*(L/R)^2 p -> dpii/pii= dp/p
;suhteellinen virhe rel_teo=sigma_p_teo/p

   rel_teo=sigma_p_teo/p
   print,'N,R/L, PII/pi dpii/pi=',N,RL,pii/!pi,rel_teo


;------------------------------------------------------------
;piirretty selostuksessa oleva kuva:
   iplot=0
   if(iplot eq 1) then begin
      nwin
      plot,[0,0],[0,0],xr=[-2.5,3.5],yr=[-2.5,3.5],xs=1,ys=1,/iso,$
           title='kolikon sade R/L='+string(rl,'(f6.2)')
      for i=-2,2,2 do begin
         oplot,i*[1,1],[-3,4],lines=1
         oplot,[-3,4],i*[1,1],lines=1
      endfor
      oplot,[0,1,1,0,0],[0,0,1,1,0],lines=2
      oplot,[0,2,2,0,0],[0,0,2,2,0],thick=2
      fii=findgen(101)/100.*2*!pi
      oplot,cos(fii)*rl,sin(fii)*rl,thick=2,lines=2
      oplot,x(hit),y(hit),psym=3
   endif

end
```

SAMA ALIOHJELMANA

pii_kolikko_f.pro

```
;--------------------------------------------------------------
  pro pii_kolikko_f,N,RL,pii,rel_teo,print=print
;--------------------------------------------------------------
;esimerkki pii:n arvioimiseksi kolikkoa heittamalla
;-------------------------------------------------------------
;Heitetaan kolikkoa, jonka halkaisija on 2R,
;ruudukolle jonka ruutujen koko on 2L  (oletetaan 2R < 2L)
;Merkitaan kolikon keskipisteen paikkaa x,y
;(mitattuna mielivaltaisesta ruudukon leikkauspisteesta)
;Milla todennakoisyydella kolikko peittaa yhden ruudukon
;leikkauskohdista = hit?
;-------------------------------------------------------------
;symmetria -> voidaan tarkastella aluetta
;             0<x<L, 0<y<L
;mikali       x^2+y^2 < R^2 -> peittaa ruudukon leikkauskohdan

;arvio todennakoisyydelle =  N_hit/N
;jossa N on kaikkien heittojen lukumaara ja N_hit 'osumien maara'

;teoreettisesti tiedetaan etta odotusarvo=
;kolikon ala/ruudun ala  = pi/4 (R/L)^2

; --> arvio pii= 4*(L/R)^2 N_hit/N
;-------------------------------------------------------------
;aliohjelma-versio, ohjeet kaytosta:
    if(n_params() le 0) then begin
       print,'pii_kolikko_f,N,RL,pii,rel_teo'
       print,'lasketaan pii:n arvo kolikkoa heittamalla'
       print,'INPUT:    '
       print,'          N= heittojen lukumaara'
       print,'          RL= R/L  rahan lapimitta/ryydun koko'
       print,'OUTPUT:   '
       print,'          PII = arvio pi:lle'
       print,'          rel_teo = teoreettinen arvio suhteelliselle virheelle'
       print,'KEYWORDS:'
       print,'          /print -> tulosta'
       return
    endif
;-------------------------------------------------------------
;   N=1000             ;heittojen lukumaara
   x=randomu(seed,N)
   y=randomu(seed,N)
;   RL=0.8                       ;kolikon sade R/L
   R2=RL^2

   hit=where((x^2+y^2) le R2, N_hit)
   pii=4./R2*N_hit/N

;teoreettinen virhearvio
;osumistodennakoisyys p= pi/4 (R/L)^2 kussakin heitossa
;Riippumattomat heitot, binomitodennakoisyys -->
;sigma_p_teo=sqrt(p(1-p)/N)

   p= !pi/4 * (RL)^2
   sigma_p_teo=sqrt(p*(1-p)/N)

;arvio pii= 4*(L/R)^2 p -> dpii/pii= dp/p
;suhteellinen virhe rel_teo=sigma_p_teo/p

   rel_teo=sigma_p_teo/p
   if(keyword_set(print)) then begin
      print,'N,R/L, PII/pi dpii/pi=',N,RL,pii/!pi,rel_teo
   endif
end
```

KUTSUTAAN ALIOHJELMAA ERI N-arvoilla

`pii_kolikko_driver.pro`

```
;------------------------------------------------------------------
program='pii_kolikko_driver'
ps=2
psdirect,program,ps                          ;avaa ps-tulostuksen
;------------------------------------------------------------------
!p.multi=[0,2,1]
  nwin

  ntab=[1d1,1d2,1d3,1d4,1d5,1d6] ;heittojen mÃďÃďrÃď yhdessÃď heittokokeessa
  k=30                           ;kutakin heittokoetta k:n sarja
  nn=n_elements(ntab)
  piitab=fltarr(nn,k)           ;taltioidaan jokaisen heittokokeen tulos
  piitab_mean=fltarr(nn)        ;keskiarvo kullekin N
  errtab_teo=fltarr(nn)         ;teoreettinen keskivirhe kullekin N
  errtab_obs=fltarr(nn)         ;havaittu keskivirhe kullekin N
  rl=1.

  plot,ntab,ntab*0.+!pi,/xlog,xtitle='N',ytitle='PII',$
       xr=[1,1e7],yr=[2,4],xs=1,title='pii_kolikko_driver.pro  R/L=1.'

  for in=0,nn-1 do begin
     n=ntab(in)
     for ik=0,k-1 do begin
        pii_kolikko_f,n,rl,pii,rel_err
        piitab(in,ik)=pii
        plots,n,pii,psym=6,syms=0.2
        if(ik eq 0) then errtab_teo(in)=rel_err*!pi
     endfor
     errtab_obs(in)=stdev(piitab(in,*)) ;otoshajonta
     piitab_mean(in)=mean(piitab(in,*)) ;keskiarvo
  endfor

  oplot,ntab,piitab_mean,lines=2
  oplot,ntab,piitab_mean+errtab_obs,lines=2
  oplot,ntab,piitab_mean-errtab_obs,lines=2
  oplot,ntab,!pi+errtab_teo,lines=0,col=2
  oplot,ntab,!pi-errtab_teo,lines=0,col=2

  nwin
   plot,ntab,errtab_teo/!pi,/xlog,/ylog,xtitle='N',ytitle='rel error',$
       xr=[1,1e7],yr=[1d-4,1],xs=1,title='pii_kolikko_driver.pro  R/L=1.'
   oplot,ntab,errtab_obs/!pi,lines=2,psym=-6
;-----------------------------------------------------------------------
psdirect,program,ps,/stop
!p.multi=0
end
```





2 Haluttua jakaumaa noudattavien satunnaismuuttujien luonti

Luennoilla on annettu menetelmä, jonka avulla voidaan luoda 1-ulotteista jakaumafunkiota $f(x)$ noudattavia satunnaismuuttujia, käyttäen välille $[0,1]$ tasaisesti jakautuneita satunnaismuuttujia. Merkitään $R_i \sim$ Tas[0,1], ja ratkaistaan R_i vastaava satunnaismuuttuja x_i asettamalla

$$\int_{-\infty}^{x_i} f(x)dx = R_i$$

Eli mikäli $f(x)$:n kertymäfunktion $F(x)$ käänteisfunktio olemassa

$$x_i = F^{-1}(R_i)$$

Miten luot seuraavia jakaumafunktioita noudattavia muuttujia (laske myös normeeraustekijat ehdosta $\int f(y)dy = 1$ yli koko arvoalueen)

a) $f(y) \sim y \qquad 0 \leq y \leq 1$

b) $f(y) \sim y^k \qquad 0 \leq y \leq 1$

c) $f(y) \sim \exp(-\lambda y) \qquad 0 \leq y$

Tarkista tulokset vertaamalla haluttuun jakaumaan (käytännässä taulukoi luotujen satunnaislukujen määrä eri jakoväleillä)

RATKAISUT:

`jakaumat.pro`

```
program='jakaumat'
ps=0
psdirect,program,ps,/color
;---------------------------------------------------------------
;luodaan jakaumafunktiot
;a) f(y)= 2y                -> F(y)=y^2
;b) f(y)= (k+1)*y^k         -> F(y)=y^(k+1)
;c) f(y)= L*exp(-L*y)       -  F(Y)=1-exp(-L*Y)
;vastaavat kertymafunktiot ja niiden kaanteisfunktiot
;a)  F(y) = y^2 = R          --> y=sqrt(R)
;b)  F(y) = y^(k+1) = R      --> y=r^(1/(k+1))
;c)  F(y) = 1-exp(-L*Y)=R    --> y = -1/L * alog(1-R)

n=1000
ns='  N='+string(n,'(i8)')
r=randomu(seed,n)

nwin
!p.multi=[0,1,3]
!p.charsize=2
;---------------------------------------------------------------------------
;a)
y=sqrt(R)
yarg=findgen(100)/100.
farg=2*yarg
histo_f,y,0,1,0.01,xx,yy                ;histo_f,x,x1,x2,dx,xx,yy
```

```
plot,yarg,farg,xtitle='y',ytitle='f(y)',title='f(y)=2y'+ns
oplot,xx,yy,psym=10,col=2
;-----------------------------------------------------------------------------
;b)
k=4
y=R^(1./(1.+k))
yarg=findgen(100)/100.
farg=(k+1)*yarg^k
histo_f,y,0,1,0.01,xx,yy
plot,yarg,farg,xtitle='y',ytitle='f(y)',title='f(y)=(k+1)*y^k, with k='+string(k,'(i4)')+ns
oplot,xx,yy,psym=10,col=2
;-----------------------------------------------------------------------------
;c)
L=2.
y=-1./L*alog(1-R)
yarg=findgen(300)/100.*L
farg=L*exp(-L*yarg)
histo_f,y,0,6,0.01,xx,yy
plot,yarg,farg,xtitle='y',ytitle='f(y)',title='f(y)=L*exp(-L*yarg), with L=2'+ns
oplot,xx,yy,psym=10,col=2
;-----------------------------------------------------------------------------
psdirect,program,ps,/color,/stop
end
```




f(y)=2y N=   10000
f(y)
2.0
1.5
1.0
0.5
0.0
0.0
0.2
0.4
0.6
0.8
1.0
y


f(y)=(k+1)*y^k, with k=   4 N=   10000
f(y)
5
4
3
2
1
0
0.0
0.2
0.4
0.6
0.8
1.0
y


f(y)=L*exp(-L*yarg), with L=2 N=   10000
f(y)
2.0
1.5
1.0
0.5
0.0
0
2
4
6
y


f(y)=2y N= 100000
f(y)
2.0
1.5
1.0
0.5
0.0
0.0
0.2
0.4
0.6
0.8
1.0
y


f(y)=(k+1)*y^k, with k=   4 N= 100000
f(y)
5
4
3
2
1
0
0.0
0.2
0.4
0.6
0.8
1.0
y


f(y)=L*exp(-L*yarg), with L=2 N= 100000
f(y)
2.0
1.5
1.0
0.5
0.0
0
2
4
6
y

3) Määrätyn integraalin laskeminen MC-menetelmällä

Kokeile luennoilla esitettyjä menetelmiä 1 ja 2 (keskiarvo menetelmä ja pinta-alan lasku) esim. integraalien

$$\int_0^\pi \sin x \; dx$$

$$\int_0^\pi \sin^2 x \; dx$$

laskemiseen. Totea miten eri menetelmien tulokset konvergoivat N:n kasvaessa.

RATKAISUT:

`mc_sini.pro`

```
program='mc_sini'
ps=0
psdirect,program,ps,/col


;---------------------------------------------
;integraali sin(x)  0 - \pi
;teoreettinen:  sijoitus -cos(x)  =   cos(0)-cos(\pi)=2
   int0=2.

   N=100
   x=randomu(seed,N)*!pi
   c=1.                          ;rajaa sin(x)
   y=randomu(seed,N)*c

;Menetelma 1 - keskiarvo
   f=sin(x)
   int1=mean(f)*!pi
;Menetelma 2 - hit or miss
   ind=where(y le sin(x),n_hit)
   int2=c*!pi*n_hit/n

;teoreettiset virhearviot
;menetelma 1:
;varianssi*N = (b-a)* integraali f^2(x) dx -(integraali f(x)dx)^2
   vari1= !pi* !pi/2. -2^2

;menetelma 2:
;varianssi*N = c*(b-a)integraali f^(x) dx -(integraali f(x)dx)^2
   vari2=1*!pi*2-2*2

   print,'N,int0,dint1,int2',n,int0,int1,int2
   print,sqrt(vari1/n),sqrt(vari2/n)

;------------------------------------------------------------
;katsotaan miten tulokset jakaantuvat kun lasketaan k kertaa
;kayttaen eri satunnaisotosta

   k=1000
   int1tab=fltarr(k)
   int2tab=fltarr(k)

   for ik=0,k-1 do begin
      N=1000
      x=randomu(seed,N)*!pi
      c=1.                       ;rajaa sin(x)
```

```
      y=randomu(seed,N)*c
;Menetelma 1 - keskiarvo
      f=sin(x)
      int1=mean(f)*!pi
;Menetelma 2 - hit or miss
      ind=where(y le sin(x),n_hit)
      int2=c*!pi*n_hit/n

      int1tab(ik)=int1
      int2tab(ik)=int2
   endfor

;plotataan tulosten jakauma + virhearviota vastaava gaussinen
;jakauma (osoittaa etta KRAL toimii hyvin!)

   nwin
   histo_f,int1tab,1.8,2.2,.01,xx,yy1
   plot,xx,yy1,psym=10,title='sin(x) integroitu 0 - pi, N=1000',/nod
   oplot,xx,yy1,psym=10,col=2,lines=0
   histo_f,int2tab,1.8,2.2,.01,xx,yy2
   oplot,xx,yy2,psym=10,col=3,lines=2

   haj1=sqrt(vari1/n)
   oplot,xx,1./(sqrt(2*!pi)*haj1)*exp(-0.5*((xx-int0)/haj1)^2),col=2
   haj2=sqrt(vari2/n)
   oplot,xx,1./(sqrt(2*!pi)*haj2)*exp(-0.5*((xx-int0)/haj2)^2),col=3,lines=2

   ff='(f6.3)'
   label_data,0.65,0.9,['Keskiarvo'+string(haj1,ff),'Hit or miss'+string(haj2,ff)],col=[2,3],lines=[0,2],len=0.075

   xyouts,1.82,13,'pylvaat=saatujen MC-tulosten jakauma',col=5
   xyouts,1.82,12,'viiva=Normaalijakauma jolla!csama keskivirhe',col=5

  psdirect,program,ps,/stop


end
```



sin(x) integroitu 0 - pi, N=1000
pylvaat=saatujen MC-tulosten jakauma
viiva=Normaalijakauma jolla
sama keskivirhe
Keskiarvo 0.031
Hit or miss 0.048
15
10
5
0
1.8
1.9
2.0
2.1
2.2